

Ecography

ECOG-03886

Wells, K., Gibson, D. I. and Clark, N. J. 2019. Global patterns in helminth host specificity: phylogenetic and functional diversity of regional host species pools matter. – Ecography doi: [10.1111/ecog.03886](https://doi.org/10.1111/ecog.03886)

Supplementary material

1 **Appendix 1.**

2

3 **Table A1.** Summary of the number of mammalian host species and regions in which the focal
 4 parasite species have been recorded.

Species	# host species	Host order	# regions	Region names
<i>Calodium hepaticum</i>	67	Rodentia (53), Carnivora (5), Primates (4), Lagomorpha (2), Artiodactyla (1), Eulipotyphla (1), Perissodactyla (1)	10	Afrotropical, Australian, Nearctic, Neotropical, Oceanian, Oriental, Palearctic, Panamanian, Saharo-Arabian, Sino-Japanese
<i>Echinococcus granulosus</i>	68	Artiodactyla (26), Diprotodontia (17), Carnivora (12), Rodentia (5), Perissodactyla (3), Primates (2), Lagomorpha (1), Pholidota (1), Proboscidea (1)	9	Afrotropical, Australian, Nearctic, Neotropical, Oriental, Palearctic, Panamanian, Saharo-Arabian, Sino-Japanese
<i>Echinococcus multilocularis</i>	52	Rodentia (32), Carnivora (10), Eulipotyphla (4), Artiodactyla (3), Primates (2), Perissodactyla (1)	4	Nearctic, Palearctic, Saharo-Arabian, Sino-Japanese

<i>Hydatigera taeniaeformis</i>	80	Rodentia (53), Carnivora (22), Eulipotyphla (2), Lagomorpha (2), Primates (1)	9	Afrotropical, Australian, Nearctic, Neotropical, Oriental, Palearctic, Panamanian, Saharo-Arabian, Sino-Japanese
<i>Hymenolepis diminuta</i>	64	Rodentia (60), Eulipotyphla (3), Primates (1)	10	Afrotropical, Australian, Nearctic, Neotropical, Oceanian, Oriental, Palearctic, Panamanian, Saharo-Arabian, Sino-Japanese
<i>Mesocestoides lineatus</i>	59	Carnivora (26), Rodentia (22), Eulipotyphla (8), Didelphimorphia (1), Lagomorpha (1), Primates (1)	5	Nearctic, Neotropical, Oriental, Palearctic, Saharo-Arabian
<i>Taenia hydatigena</i>	60	Artiodactyla (29), Carnivora (14), Rodentia (11), Primates (3), Diprotodontia (1), Lagomorpha, (1), Perissodactyla (1)	8	Afrotropical, Australian, Nearctic, Neotropical, Oriental, Palearctic, Saharo-Arabian, Sino-Japanese

<i>Trichinella spiralis</i>	61	Carnivora (36), Rodentia (13), Artiodactyla (5), Eulipotyphla (3), Didelphimorphia (2), Perissodactyla (1), Primates (1)	8	Afrotropical, Australian, Nearctic, Neotropical, Oriental, Palaeartic, Saharo-Arabian, Sino-Japanese
<i>Versteria mustelae</i>	74	Rodentia (53), Carnivora (11), Eulipotyphla (7), Artiodactyla (2), Primates (1)	5	Nearctic, Oriental, Palaeartic, Saharo-Arabian, Sino-Japanese

5
6
7
8
9
10
11
12
13
14
15
16
17
18
19
20

21 **Box A1.** Model code for estimating the difference in phylogenetic and functional diversity of
 22 region host species assemblages from regional species pools.

23 Data objects are *#N*: number of data points in vectors of pairwise phylogenetic and ecological
 24 distances between observed/expected host species; *#NRegion*: number of zoogeographic
 25 regions; *#phyldist*: vector of pairwise phylogenetic distances between observed/expected host
 26 species; *#ecoldist*: vector of pairwise ecological distances between observed/expected host
 27 species; *#type*: binary indicator variable if pair of host species has been observed or is
 28 expected according to samples mammalian species pool.

```

model {
  for(i in 1:N){
    phyldist[i] ~ dnorm(phyll[i], tau.phyl)
    ecoldist[i] ~ dnorm(ecol[i], tau.ecol)

    phyll[i] <- mu.phyll[region[i]] + b.phyll[region[i]] * equals(type[i],2)
    ecol[i] <- mu.ecol[region[i]] + b.ecol[region[i]] * equals(type[i],2)
  }

  for(r in 1:NRegion){
    mu.phyll[r] ~ dnorm(muHyp.phyll, tau.mu.phyll)
    mu.ecol[r] ~ dnorm(muHyp.ecol, tau.mu.ecol)

    b.phyll[r] ~ dnorm(bHyp.phyll, tau.bphyll)
    b.ecol[r] ~ dnorm(bHyp.ecol, tau.becol)

    inf.phyll[r] <- mu.phyll[r] + b.phyll[r]
    inf.ecol[r] <- mu.ecol[r] + b.ecol[r]
  }

  tau.phyll <- pow(sd.phyll, -2); sd.phyll ~ dexp(0.5)
  tau.ecol <- pow(sd.ecol, -2); sd.ecol ~ dexp(0.5)
  tau.mu.phyll <- pow(sd.mu.phyll, -2); sd.mu.phyll ~ dexp(0.5)
  tau.mu.ecol <- pow(sd.mu.ecol, -2); sd.mu.ecol ~ dexp(0.5)
  tau.bphyll <- pow(sd.bphyll, -2); sd.bphyll ~ dexp(0.5)
  tau.becol <- pow(sd.becol, -2); sd.becol ~ dexp(0.5)

  muHyp.phyll ~ dnorm(0, 0.01)
  muHyp.ecol ~ dnorm(0, 0.01)
  bHyp.phyll ~ dnorm(0, 0.01)
  bHyp.ecol ~ dnorm(0, 0.01)
}

```

29

30

31

32

33

34

35 **Box A2.** Model code for probabilistic estimates of average association rates within regional
 36 mammalian species pools. Data objects are *#NSp*: number of focal parasite species;
 37 *#NRegion*: number of zoogeographic regions; *#ninf_noncarn*: number of observed non-
 38 carnivoran host species in regional species pool; *#nsamp_noncarn*: number of expected
 39 (sampled) non-carnivoran host species in regional species pool; *#ninf_carn*: number of
 40 observed carnivoran host species in regional species pool; *#nsamp_carn*: number of expected
 41 (sampled) carnivoran host species in regional species pool.

42

```

model {
  for(i in 1:NSp){
    for(r in 1:NRegion){
      ninf_noncarn[i,r] ~ dbin(phi_noncarn[i,r], nsamp_noncarn[i,r])
      ninf_carn[i,r] ~ dbin(phi_carn[i,r], nsamp_carn[i,r])

      logit(phi_noncarn[i,r]) <- lphi_noncarn[i,r]
      lphi_noncarn[i,r] ~ dnorm(mu.phi_noncarn[i], tau.phi_noncarn)

      logit(phi_carn[i,r]) <- lphi_carn[i,r]
      lphi_carn[i,r] ~ dnorm(mu.phi_carn[i], tau.phi_carn)
    }
    mu.phi_noncarn[i] ~ dnorm(0, 0.01)
    mu.phi_carn[i] ~ dnorm(0, 0.01)
  }
  tau.phi_noncarn <- pow(sd.phi_noncarn, -2); sd.phi_noncarn ~ dexp(0.5)
  tau.phi_carn <- pow(sd.phi_carn, -2); sd.phi_carn ~ dexp(0.5)
}

```

43

44

45

46

47

48

49

50

51

52

53

54

55

56

57 **Box A3.** Model code for phylogenetic generalized linear model to explore the relative impact
58 of different host traits on infection probability with a parasite.

59 Data objects are *#N*: number of data point in data vector *Y*; *#Y*: binary vector of whether a
60 sampled host species is infected (observed host = 1) or not (expected host = 0) with the focal
61 parasite; *#X*: matrix of all trait variables for each host species in the data vector *Y*; *#ntrait*:
62 number of trait variables; *#Xmean*: mean of trait variables for various orders; *#XSD*: 1 SD of
63 trait variables for various orders; *#hostorder*: mammalian order for each host species
64 recorded in data vector *Y*; *#phyldist*: phylogenetic distance matrix between all pairs of host
65 species.

66

```

model {
  for(i in 1:N){
    Y[i] ~ dbern(phi[i])
    logit(phi[i]) <- mu + inprod(X.sc[i,],beta[]) + W.phylo[i]
    muW.phylo[i] <- 0
  }

  ## Prior
  for(x in 1:ntrait){
    beta[x]~ dnorm(0, tau.b[x])
    tau.b[x] <- pow(sd.b[x], -2); sd.b[x] ~ dexp(1)
  }

  W.phylo[1:N] ~ dnorm(muW.phylo[], Omega[,])
  tau.w. <- pow(sd.w, -2); sd.w ~ dexp(0.5)
  gamma ~ dexp(1)
  mu ~ dnorm(0, 0.001)

  for(i in 1:N){
    for(j in 1:N){
      H[i,j] <- (1/tau.w) * exp(-gamma * pow(phyldist[i,j],2))
    }
  }
  Omega [1:N,1:N] <- inverse(H[1:N,1:N])


  ## Impute missing covariate values according to order-specific values and scale
  for(i in 1:N){
    for(x in 1:ntrait){
      X[i,x] ~ dnorm(Xmean[hostorder[i],x], XSD[hostorder[i],x])
      X.sc[i,x] <- (X[i,x] - cov.mean[x])/ cov.sd[x]
    }
  }
  for(x in 1:ntrait){
    cov.mean[x] <- mean(X[1:N,x])
    cov.sd[x] <- sd(X[1:N,x])
  }
}

```

67

68

69



70

71 **Figure A1.** Relative effect sizes of different covariates on the association probability of non-
 72 carnivoran mammalian host species with some of the most invasive and globally spread
 73 helminth species for a model that does *NOT* account for underlying phylogenetic
 74 relationships among host species. See Figure 4 in main text for model output for an
 75 equivalent phylogenetic generalised linear model that account for such phylogenetic
 76 relationships. Note that Carnivora were not included in the analysis to focus on species that
 77 most likely serve as intermediate host for those with indirect transmission cycles (except
 78 *Hymenolepis diminuta*, for which mammals are definitive hosts and arthropods are
 79 intermediate hosts).

80